Unbeabsichtigtes Beenden von Access verhindern, Teil 1
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

Viele Datenbanken sollen vom Anwender nur über eine spezielle Schaltfläche verlassen werden, weil beispielsweise beim Beenden noch bestimmte Aktionen wie der Export von Daten oder der Druck von Auswertungen ausgeführt werden müssen. Dazu blenden Sie in der Regel die Standrad-Menüs aus und verwenden eigene, per Menümakro realisierte Menüs, die keinen Datei-Beenden-Eintrag enthalten. Allerdings kann Access immer noch über das Schließen-Symbol in der Titelleiste unkontrolliert vom Anwender beendet und so das Ausführen von "Beenden-Aktionen" unterlaufen werden.

Dieses Problem lässt sich jedoch mit einigen API-Funktionen lösen: Ausschlaggebend für die Möglichkeit, eine Anwendung per Doppelklick auf das Schließen-Symbol oder Anwahl des Schließen-Eintrages im Anwendungsmenü beenden zu können, ist das Vorhandensein des Anwendungsmenüs. Wenn man dieses Menü löscht, führt einerseits ein Doppelklick darauf ins Leere, so dass die Datenbank auf diesem Wege nicht mehr verlassen werden kann und andererseits reagiert das Schließen-Symbol in der Titelleiste ebenfalls nicht mehr.

Um das Anwendungsmenü von Access zu entfernen, benötigen Sie zunächst eine kleine Funktion "KillClose", die mit Hilfe von API-Funktionen das Anwendungsmenü löscht sowie ein AutoExec-Makro, das diese Funktion beim Öffnen der Datenbank automatisch aufruft:

· Kopieren Sie die folgenden Deklarationen und die Funktion "KillClose()" in ein neues Modul, das Sie beispielsweise unter "modKillClose" speichern:

Const MF_BYPOSITION = &H400&

Declare Function GetSystemMenu _
        Lib "user32" ( _
        ByVal hwnd As Long, _
        ByVal bRevert As Long) As Long

Declare Function RemoveMenu _
        Lib "user32" ( _
        ByVal hMenu As Long, _
        ByVal nPosition As Long, _
        ByVal wFlags As Long) As Long

Declare Function DrawMenuBar _
        Lib "user32" ( _
        ByVal hWind As Long) As Long


Function KillClose()
  Dim hMenu As Long
  Dim I&, R As Variant

  hMenu = GetSystemMenu(Application.hWndAccessApp, 0)
  For I = 6 To 0 Step -1
    R = RemoveMenu(hMenu, I, MF_BYPOSITION)
  Next

  DrawMenuBar hMenu
  Application.RefreshTitleBar

End Function
· Legen Sie ein Makro "AutoExec" an bzw. erweitern Sie ein vorhandenes AutoExec-Makro um die Aktion "AusführenCode" und geben Sie für den Parameter "Funktion" den Ausdruck "KillClose()" ein.

Die Funktion "KillClose()" wird nun automatisch beim Öffnen der Datenbank aufgerufen und eliminiert die Möglichkeiten, Access über das Anwendungsmenü oder das Schließen-Symbol zu beenden. Auch ein Rechtsklick auf das Symbol von Access in der Windows-Taskleiste und der Versuch, es hier über Schließen zu beenden, bleibt wirkungslos.

Zur Arbeitsweise der Funktion "KillClose()": "GetSystemMenu" liefert zunächst anhand des Window Handles "hWndAccApp" eine Referenz "hMenu" auf das Anwendungsmenü von Access. In einer Schleife löschen wir dann über "RemoveMenu" die darin enthaltenen sechs Einträge. Ein Aufruf von "DrawMenuBar" und "RefreshTitleBar" aktualisiert die Titelleiste, so dass das Schließen-Symbol deaktiviert angezeigt wird. Ein Doppelklick auf das Anwendungsmenü bzw. ein Klick auf das Schließen-Symbol beendet Access nicht mehr.

Damit sind nun zwar die "anklickbaren" Möglichkeiten unterbunden, aber es bleibt noch der Weg, Access per Tastenkombination Alt+F4 unkontrolliert zu beenden. Wie Sie dieses Problem lösen, erläutern wir nächste Woche im zweiten Teil dieses Tipps.
16.11.2007
C:\_CD_brennen_div\Autocad\2006\Installationsanleitung AutoCad LT 2006 mit Menu mj.doc
2 von 2

